
PRÁCTICAS DE VISUAL BASIC APLICADO AL CÁLCULO ACTUARIAL

Eva Boj del Val

M^a Mercè Claramunt Bielsa

Isabel Morillo López

DEPARTAMENTO DE MATEMÁTICA ECONÓMICA, FINANCIERA Y ACTUARIAL

Facultad de Ciencias Económicas y Empresariales

Universidad de Barcelona

TÍTULO: Prácticas de Visual Basic aplicado al Cálculo Actuarial

© **AUTORAS:** Eva Boj del Val

M^a Mercè Claramunt Bielsa

Isabel Morillo López

EDITORAS: las autoras

1^a edición

Fecha de publicación: mayo de 2004

ÍNDICE

1. Introducción	3
2. Práctica 1: Rentas financieras	4
2.1. Enunciado	4
2.2. Carátula	5
2.3. Controles	6
2.4. Código	8
2.5. Ejemplos numéricos	11
3. Práctica 2: Rentas de supervivencia	13
3.1. Enunciado	13
3.2. Carátula	14
3.3. Controles	15
3.4. Código	17
3.5. Ejemplos numéricos	21
4. Práctica 3: Seguro mixto	22
4.1. Enunciado	22
4.2. Carátula	23
4.3. Controles	24
4.4. Código	26
4.5. Ejemplos numéricos	29

1. Introducción

Esta publicación incluye tres ejemplos de aplicación del lenguaje de programación Visual Basic al cálculo financiero y actuarial.


Constituye el material de las clases prácticas utilizado en el curso de extensión universitaria *Visual Basic aplicat al càlcul actuarial* organizado por el Departamento de matemática económica, financiera y actuarial de la Universidad de Barcelona y que viene realizándose desde el curso 1999-2000. El objetivo de este material es, pues, docente; por ello los programas aquí incluidos están realizados de manera didáctica con la finalidad de aplicar los conocimientos estudiados en la parte teórica de este curso introductorio a la programación en Visual Basic.

Organizamos un proyecto de nombre “Cálculo Actuarial” dentro del cual incluimos tres formularios que se corresponden con cada práctica. La práctica 1 hace referencia a rentas financieras, la práctica 2 a rentas de supervivencia y la práctica 3 a seguros mixtos.

PROYECTO: Cálculo Actuarial.vbp

FORMULARIOS:

- Práctica 1.frm (Rentas financieras)
- Práctica 2.frm (Rentas de supervivencia)
- Práctica 3.frm (Seguro mixto)



2. Práctica 1: Rentas financieras

2.1. Enunciado

Realizar un programa que calcule el valor actual de una renta financiera anual, en el que podamos elegir el tipo de cuantía: constante, variable linealmente o variable geoméricamente.

Datos a entrar por pantalla:

1. Tanto de valoración en tanto por uno (se solicita el efectivo anual)
2. Respecto a la renta:
 - Diferimiento (en años)
 - Si es anticipada o vencida
 - Si es temporal o perpetua.
En el caso de ser temporal: Número de términos
 - Cuantía de la renta:
 - Constante: cuantía en euros
 - Variable linealmente: cuantía del primer término en euros y razón de variación lineal
 - Variable geoméricamente: cuantía del primer término en euros y razón de variación geométrica

Resultados a presentar por pantalla:

- Valor Actual de la Renta
- Valor Final de la Renta

El formulario debe incluir los botones: CALCULAR, OTRO y SALIR.

2.2. Carátula


Sin ejecutar

The screenshot shows a window titled "Valor Actual Renta Anual" with a dotted background. It is divided into two main sections: "Datos" on the left and "Resultados" on the right. The "Datos" section contains several input fields and radio buttons: "Interés Efectivo Anual (en tanto por uno)" with a text box; "Diferimiento en años" with a text box; "Inicio pagos" with radio buttons for "Por anticipado" and "Por vencido"; "Temporalidad" with radio buttons for "Perpetua" and "Temporal"; "Número de términos anuales" with a text box; "Cuantía de los términos" with a dropdown menu for "Tipo de variación de la renta:"; "Cuantía del primer término en euros" with a text box; and "Razón de variación" with a text box. The "Resultados" section has two empty text boxes for "Valor Actual en euros" and "Valor Final en euros". Below these are three buttons: "CALCULAR", "OTRO", and "SALIR".

Ejecutada

This screenshot is identical to the one above, showing the "Valor Actual Renta Anual" window. The layout, labels, and input fields are the same. The only difference is the presence of a mouse cursor pointing at the dropdown menu for "Tipo de variación de la renta:" in the "Cuantía de los términos" section.

Combo Box



2.3. Controles

CONTROL	(NOMBRE)	CAPTION	TEXT
Formulario	frmPractica1	Valor Actual Renta Anual	
Frame	frDatos	Datos	
Frame	frResultados	Resultados	
Frame	frCuantia	Cuantía de los términos	
Command Button	cmdCALCULAR	CALCULAR	
Command Button	cmdOTRO	OTRO	
Command Button	cmdSALIR	SALIR	
Label	lblInteres	Interés Efectivo Anual (en tanto por uno)	
Text Box	txtInteres		En blanco
Label	lblDiferimiento	Diferimiento en años	
Text Box	txtDiferimiento		En blanco

Frame	frInicioPagos	Inicio pagos	
Option Button	opbAnticipada	Por anticipado	
Option Button	opbVencida	Por vencido	
Frame	frTemporalidad	Temporalidad	
Option Button	opbPerpetua	Perpetua	
Option Button	opbTemporal	Temporal	
Label	lblTerminos	Número de términos anuales	
Text Box	txtTerminos		En blanco
Label	lblCuantia	Cuantía del primer término en euros	
Text Box	txtCuantia		En blanco
Label	lblValorActual	Valor Actual en euros	
Label	lblValorFinal	Valor Final en euros	
Label	lblResultado1	En blanco	
Label	lblResultado2	En blanco	
Label	lblTipoRenta	Tipo de variación de la renta:	
Label	lblRazon	Razón de variación	
Text Box	txtRazon		En blanco

Combo Box	cmbCuantia		En blanco
-----------	------------	--	-----------

2.4. Código

'Definición de Variables en (General) (Declaraciones)

```
Dim Razon As Double
Dim Interes As Double
Dim NuevolInteres As String
Dim Diferimiento As Double
Dim Terminos As Integer
Dim Cuantia As Double
Dim ValorActual As Double
Dim ValorFinal As Double
```

```
Private Sub cmbCuantia_Change()
    If cmbCuantia.Text = "Lineal" Or cmbCuantia.Text = "Geométrica" Then
        lblCuantia.Visible = True
        txtCuantia.Visible = True
        lblRazon.Visible = True
        txtRazon.Visible = True
    ElseIf cmbCuantia.Text = "Constante" Then
        lblCuantia.Visible = True
        txtCuantia.Visible = True
        lblRazon.Visible = False
        txtRazon.Visible = False
    Else
        lblCuantia.Visible = False
        txtCuantia.Visible = False
        lblRazon.Visible = False
        txtRazon.Visible = False
    End If
End Sub
```

```
Private Sub cmbCuantia_Click()
    lblCuantia.Visible = True
    txtCuantia.Visible = True
    If cmbCuantia.ListIndex = 1 Or cmbCuantia.ListIndex = 2 Then
        lblRazon.Visible = True
        txtRazon.Visible = True
    Else
        lblRazon.Visible = False
        txtRazon.Visible = False
    End If
End Sub
```

```
Private Sub CmdCalcular_Click()
```

```
'Obtener Información
Diferimiento = Val(txtDiferimiento.Text)
Terminos = Val(txtTerminos.Text)
```

```
Cuantia = Val(txtCuantia.Text)
Razon = Val(txtRazon.Text)
Interes = Val(txtInteres.Text)
' InputBox para controlar que el interés sea positivo
If Interes < 0 Then
    NuevolInteres = InputBox("El Interés debe ser estrictamente positivo." & vbCrLf & "Introduzca de nuevo el
Interés.", "ERROR")
    If NuevolInteres = "" Then
        cmdOTRO_Click
        Exit Sub
    End If
    Interes = Val(NuevolInteres)
    txtInteres.Text = Str(Interes)
End If

'Cálculos
If opbPerpetua.Value = True Then
    If (cmbCuantia.ListIndex = 0 Or cmbCuantia.Text = "Constante") Then
        ValorActual = Cuantia * (1 / Interes) * ((1 + Interes) ^ (-Diferimiento - opbAnticipada.Value))
    ElseIf (cmbCuantia.ListIndex = 1 Or cmbCuantia.Text = "Lineal") Then
        ValorActual = ((Cuantia / Interes) + (Razon / (Interes ^ 2))) * ((1 + Interes) ^ (-Diferimiento -
opbAnticipada.Value))
    ElseIf (cmbCuantia.ListIndex = 2 Or cmbCuantia.Text = "Geométrica") Then
        'Control sobre la razón
        If Razon >= 1 + Interes Then
            Res = MsgBox("La Razón debe ser menor que 1+Interés." & vbCrLf & "Introduzca de nuevo la Razón.",
vbOKCancel + vbExclamation + vbDefaultButton1, "ERROR")
            If Res = vbOK Then
                txtRazon.Text = ""
                txtRazon.SetFocus
                Exit Sub
            Else
                cmdOTRO_Click
                Exit Sub
            End If
        End If
        ValorActual = (Cuantia / (1 + Interes - Razon)) * ((1 + Interes) ^ (-Diferimiento - opbAnticipada.Value))
    End If
Else
    ValorActual = 0
    i = Diferimiento + opbAnticipada.Value
    If (cmbCuantia.ListIndex = 0 Or cmbCuantia.Text = "Constante") Then
        For J = 1 To Terminos
            ValorActual = ValorActual + Cuantia * ((1 + Interes) ^ (-(J + i)))
        Next J
    ElseIf (cmbCuantia.ListIndex = 1 Or cmbCuantia.Text = "Lineal") Then
        For J = 1 To Terminos
            ValorActual = ValorActual + (Cuantia + (J - 1) * Razon) * ((1 + Interes) ^ (-(J + i)))
        Next J
    ElseIf (cmbCuantia.ListIndex = 2 Or cmbCuantia.Text = "Geométrica") Then
        For J = 1 To Terminos
            ValorActual = ValorActual + (Cuantia * (Razon ^ (J - 1))) * ((1 + Interes) ^ (-(J + i)))
        Next J
    End If
    ValorFinal = ValorActual * (1 + Interes) ^ (Diferimiento + Terminos)
End If
```

'Salida de Resultados

```
IblResultado1.Caption = Str$(ValorActual)
IblResultado2.Caption = Str$(ValorFinal)
```

End Sub

```
Private Sub cmdOTRO_Click()
    txtInteres.Text = " "
    txtDiferimiento.Text = " "
    txtTerminos.Text = " "
    IblTerminos.Visible = False
    txtTerminos.Visible = False
    txtCuantia.Text = " "
    IblCuantia.Visible = False
    txtCuantia.Visible = False
    IblRazon.Visible = False
    txtRazon.Visible = False
    txtRazon.Text = " "
    opbAnticipada.Value = False
    opbVencida.Value = False
    opbPerpetua.Value = False
    opbTemporal.Value = False
    IblResultado1.Caption = " "
    IblResultado2.Caption = " "
    IblResultado2.Visible = True
    IblValorFinal.Visible = True
    'cmbCuantia.ListIndex = -1
    cmbCuantia.Text = " "
    txtInteres.SetFocus
End Sub
```

```
Private Sub CmdSalir_Click()
    End 'Sale del programa
End Sub
```

```
Private Sub opbPerpetua_Click()
    IblResultado2.Visible = False
    IblValorFinal.Visible = False
    IblTerminos.Visible = False
    txtTerminos.Visible = False
End Sub
```

```
Private Sub opbTemporal_Click()
    IblTerminos.Visible = True
    txtTerminos.Visible = True
    IblResultado2.Visible = True
    IblValorFinal.Visible = True
End Sub
```

2.5. Ejemplos numéricos

Rentas constantes

- Calcular el valor actual y el valor final de una renta financiera anual constante de cuantía 3.000 € de 10 términos y diferida 2 años, valorada al 6% efectivo anual.

Solución:

Si la renta es anticipada:

Valor Actual = 20.830,43 € Valor Final = 41.914,93 €

Si la renta es vencida:

Valor Actual = 19.651,35 € Valor Final = 39.542,38 €

- Calcular el valor actual de una renta financiera perpetua anual constante de cuantía 900 € valorada al 1% efectivo anual.

Solución:

Si la renta es anticipada:

Valor Actual = 90.900 €

Si la renta es vencida:

Valor Actual = 90.000 €

Rentas lineales

- Calcular el valor actual y el valor final de una renta financiera anual creciente 300 € anuales con una cuantía inicial de 900 €, de 20 términos y diferida 4 años y medio, valorada al 2% efectivo anual.

Solución:

Si la renta es anticipada:

Valor Actual = 54.206,15 € Valor Final = 88.054,74 €

Si la renta es vencida:

Valor Actual = 53.143,28 € Valor Final = 86.328,18 €

- Calcular ahora el valor actual de la renta anterior, suponiendo que en lugar de ser temporal sea perpetua.

Solución:

Si la renta es anticipada:

Valor Actual = 741.765,32 €

Si la renta es vencida:

Valor Actual = 727.220,91 €

Rentas geométricas

- Calcular el valor actual y el valor final de una renta financiera anual creciente en un 2% anual acumulativo, de cuantía inicial 600 €, de 15 términos y diferida medio años, valorada al 3% efectivo anual.

Solución:

Si la renta es anticipada:

Valor Actual = 8.289,92 €

Valor Final = 13.107,72 €

Si la renta es vencida:

Valor Actual = 8.048,46 €

Valor Final = 12.725,94 €

- Calcular ahora el valor actual de la renta anterior, suponiendo que en lugar de ser temporal sea perpetua.

Solución:

Si la renta es anticipada:

Valor Actual = 60.893,95 €

Si la renta es vencida:

Valor Actual = 59.119,76 €

3. Práctica 2: Rentas de supervivencia

3.1. Enunciado

Realizar un programa que calcule el valor actual actuarial de una renta de supervivencia anual.

Datos a entrar por pantalla:

1. Respecto a la renta:

- Diferimiento (en años)
- Si es anticipada o vencida
- Si es temporal o vitalicia.

En el caso de ser temporal: Número de términos

- Cuantía de la renta:
 - Constante: cuantía en euros
 - Variable linealmente: cuantía del primer término en euros y razón de variación lineal
 - Variable geoméricamente: cuantía del primer término en euros y razón de variación geométrica

2. Respecto al asegurado:

- Fecha de nacimiento
- Fecha de valoración

3. Bases Técnicas:

- Tanto de valoración en tanto por uno (se solicita el efectivo anual)
- Tablas de mortalidad a utilizar: Tablas Generacionales PERM/2000-P

Resultado a presentar por pantalla:

- Valor Actual Actuarial de la Renta

El formulario debe incluir los botones: CALCULAR, OTRO y SALIR.

3.2. Carátula

Sin ejecutar

Valor Actual Actuarial de una Renta de Supervivencia Anual

Datos de la Renta

Diferimiento en años

Inicio pagos

Por anticipado

Por vencido

Temporalidad

Vitalicia

Temporal

Número de términos anuales

Cuantía de los términos

Tipo de variación de la renta:

Cuantía del primer término en euros

Razón de variación

Datos del Asegurado

Fecha de nacimiento (dd/mm/aaaa)

Fecha de valoración (dd/mm/aaaa)

Bases Técnicas

Interés Efectivo Anual (en tanto por uno)

Tablas Generacionales PERM/2000-P

Tablas generacionales

Valor Actual Actuarial en euros

Ejecutada

Valor Actual Actuarial de una Renta de Supervivencia Anual

Datos de la Renta

Diferimiento en años

Inicio pagos

Por anticipado

Por vencido

Temporalidad

Vitalicia

Temporal

Cuantía de los términos

Tipo de variación de la renta:

Datos del Asegurado

Fecha de nacimiento (dd/mm/aaaa)

Fecha de valoración (dd/mm/aaaa)


Bases Técnicas

Interés Efectivo Anual (en tanto por uno)

Tablas Generacionales PERM/2000-P

Valor Actual Actuarial en euros

List Box



3.3. Controles

CONTROL	(NOMBRE)	CAPTION	TEXT
Formulario	frmPractica2	Valor Actual Actuarial de una Renta de Supervivencia Anual	
Frame	frDatosRenta	Datos de la Renta	
Frame	frDatosAsegurado	Datos del Asegurado	
Frame	frBasesTecnicas	Bases Técnicas	
Label	lblDiferimiento	Diferimiento en años	
Text Box	txtDiferimiento		En blanco
Frame	frInicioPagos	Inicio pagos	
Option Button	opbAnticipada	Por anticipado	
Option Button	opbVencida	Por vencido	
Frame	frTemporalidad	Temporalidad	
Option Button	opbVitalicia	Vitalicia	
Option Button	opbTemporal	Temporal	

Label	lblTerminos	Número de términos anuales	
Text Box	txtTerminos		En blanco
Frame	frCuantia	Cuantía de los términos	
Label	lblTipoRenta	Tipo de variación de la renta:	
List Box	lsbCuantia		
Label	lblCuantia	Cuantía del primer término en euros	
Text Box	txtCuantia		En blanco
Label	lblRazon	Razón de variación	
Text Box	txtRazon		En blanco
Label	lblFechaNacimiento	Fecha de Nacimiento (dd/mm/aaaa)	
Text Box	txtFechaNacimiento		En blanco
Label	lblFechaValoracion	Fecha de Valoración (dd/mm/aaaa)	
Text Box	txtFechaValoracion		En blanco
Label	lblInteres	Interés Efectivo Anual (en tanto por uno)	
Text Box	txtInteres		En blanco
Label	lblTablas	Tablas Generacionales PERM/2000-P	
Data	dataTablas	Tablas Generacionales	

Command Button	cmdCALCULAR	CALCULAR	
Command Button	cmdOTRO	OTRO	
Command Button	cmdSALIR	SALIR	
Label	lblValorActualActuarial	Valor Actual Actuarial en euros	
Label	lblResultado	En blanco	

3.4. Código

'Definición de Variables en (General) (Declaraciones)

```
Dim Razon As Double
Dim Interes As Double
Dim NuevoInteres As String
Dim NuevoDiferimiento As String
Dim NuevoTerminos As String
Dim NuevaCuantia As String
Dim Diferimiento As Double
Dim Terminos As Integer
Dim Cuantia As Double
Dim ValorActualActuarial As Double
Dim EdadActuarial As Integer
Dim f1 As Date
Dim f2 As Date
Dim L(200) As Double
Dim AñoNacimiento As Integer
```

```
Private Sub IsbCuantia_Click()
    lblCuantia.Visible = True
    txtCuantia.Visible = True
    If IsbCuantia.ListIndex = 1 Or IsbCuantia.ListIndex = 2 Then
        lblRazon.Visible = True
        txtRazon.Visible = True
    Else
        lblRazon.Visible = False
        txtRazon.Visible = False
    End If
End Sub
```

```
Private Sub CmdCalcular_Click()

'Obtener Información
Interes = Val(txtInteres.Text)
If Interes < 0 Then
    NuevoInteres = InputBox("El Interés debe ser estrictamente positivo." & vbCrLf & "Introduzca de nuevo el Interés.", "ERROR")
    If NuevoInteres = "" Then
        cmdOTRO_Click
        Exit Sub
    End If
    Interes = Val(NuevoInteres)
    txtInteres.Text = Str(Interes)
End If
Diferimiento = Val(txtDiferimiento.Text)
If (Diferimiento - Int(Diferimiento)) > 0 Or Diferimiento < 0 Then
    NuevoDiferimiento = InputBox("El Diferimiento debe ser en años enteros y positivo" & vbCrLf & "Introduzca de nuevo el Diferimiento.", "ERROR")
    If NuevoDiferimiento = "" Then
        cmdOTRO_Click
        Exit Sub
    End If
    Diferimiento = Val(NuevoDiferimiento)
    txtDiferimiento.Text = Str(Diferimiento)
End If
Terminos = Val(txtTerminos.Text)
If (Terminos - Int(Terminos)) > 0 Or Terminos < 0 Then
    NuevoTerminos = InputBox("El número de términos debe ser entero y positivo" & vbCrLf & "Introduzca de nuevo el número de términos.", "ERROR")
    If NuevoTerminos = "" Then
        cmdOTRO_Click
        Exit Sub
    End If
    Terminos = Val(NuevoTerminos)
    txtTerminos.Text = Str(Terminos)
End If
Cuantia = Val(txtCuantia.Text)
If Cuantia < 0 Then
    NuevaCuantia = InputBox("La Cuantía debe ser positiva" & vbCrLf & "Introduzca de nuevo la Cuantía.", "ERROR")
    If NuevaCuantia = "" Then
        cmdOTRO_Click
        Exit Sub
    End If
    Cuantia = Val(NuevaCuantia)
    txtCuantia.Text = Str(Cuantia)
End If
Razon = Val(txtRazon.Text)
f1 = txtFechaNacimiento.Text
If Year(f1) < 1885 Or Year(f1) > 2001 Then
    Res = MsgBox("La Fecha de nacimiento es incorrecta." & vbCrLf & "Introduzca de nuevo la fecha.", vbOKCancel + vbExclamation + vbDefaultButton1, "ERROR")
    If Res = vbOK Then
        txtFechaNacimiento.Text = " "
        txtFechaNacimiento.SetFocus
        Exit Sub
    Else
        cmdOTRO_Click
    End If
End If
```

```
Exit Sub
End If
End If
f2 = txtFechaValoracion.Text
If Year(f2) <= Year(f1) Then
    Res = MsgBox("La Fecha de valoración es incorrecta." & vbCrLf & "Introduzca de nuevo la fecha.",
vbOKCancel + vbExclamation + vbDefaultButton1, "ERROR")
    If Res = vbOK Then
        txtFechaValoracion.Text = " "
        txtFechaValoracion.SetFocus
        Exit Sub
    Else
        cmdOTRO_Click
        Exit Sub
    End If
End If

'Cálculo de la edad Actuarial
EdadActuarial = Int(((f2 - f1) / 365.25) + 0.5)

'Lectura de la tabla generacional
AñoNacimiento = Year(f1)

campo = Trim(AñoNacimiento)
dataTablas.Recordset.MoveFirst
For i = 0 To 116
    L(i) = dataTablas.Recordset.Fields(campo)
    dataTablas.Recordset.MoveNext
Next i

'Cálculo del Valor Actual Actuarial

If opbVencida.Value = True Then
    Diferimiento = Diferimiento + 1
End If

If opbVitalicia.Value = True Then
    Terminos = 116 - EdadActuarial - Diferimiento
End If

ValorActualActuarial = 0

If IsbCuantia.ListIndex = 0 Then
    For t = Diferimiento To Terminos + Diferimiento - 1
        ValorActualActuarial = ValorActualActuarial + Cuantia * ((1 + Interes) ^ (-t)) * (L(EdadActuarial + t) /
L(EdadActuarial))
    Next t
ElseIf IsbCuantia.ListIndex = 1 Then
    For t = Diferimiento To Terminos + Diferimiento - 1
        ValorActualActuarial = ValorActualActuarial + (Cuantia * (Razon ^ (t - Diferimiento))) * ((1 + Interes) ^ (-t)) *
(L(EdadActuarial + t) / L(EdadActuarial))
    Next t
Else
    For t = Diferimiento To Terminos + Diferimiento - 1
        ValorActualActuarial = ValorActualActuarial + (Cuantia + (Razon * (t - Diferimiento))) * ((1 + Interes) ^ (-t)) *
(L(EdadActuarial + t) / L(EdadActuarial))
    Next t
End If
```

End If

'Salida de Resultados

IblResultado.Caption = Str\$(ValorActualActuarial)

End Sub

Private Sub cmdOTRO_Click()

txtInteres.Text = " "

txtDiferimiento.Text = " "

txtTerminos.Text = " "

IblResultado.Caption = " "

opbAnticipada.Value = False

opbVencida.Value = False

opbVitalicia.Value = False

opbTemporal.Value = False

IblTerminos.Visible = False

txtTerminos.Visible = False

IsbCuantia.ListIndex = 0

IsbCuantia.ListIndex = -1

txtCuantia.Text = " "

IblCuantia.Visible = False

txtCuantia.Visible = False

IblRazon.Visible = False

txtRazon.Visible = False

txtRazon.Text = " "

txtFechaNacimiento = " "

txtFechaValoracion = " "

txtDiferimiento.SetFocus

End Sub

Private Sub CmdSalir_Click()

End 'Sale del programa

End Sub

Private Sub opbTemporal_Click()

IblTerminos.Visible = True

txtTerminos.Visible = True

End Sub

Private Sub opbVitalicia_Click()

IblTerminos.Visible = False

txtTerminos.Visible = False

End Sub

3.5. Ejemplos numéricos

- Calcular la prima pura única de las siguientes operaciones actuariales realizadas sobre un asegurado con fecha de nacimiento 01/05/1966.

Bases técnicas: tanto de interés técnico 3% efectivo anual; tabla de mortalidad PERM/2000-P.

Fecha de valoración: 31/03/2004.

1. Renta pagadera mientras viva el asegurado, anual prepagable, de cuantía 300 € al año, inmediata y temporal 10 años.

Solución: 2.620,20 €

2. Renta pagadera mientras viva el asegurado y como máximo hasta que alcance los 65 años de edad, anual, vencida, diferida 10 años y de cuantía inicial 6.000 € creciendo esta cantidad anualmente en 300 €.

Solución: 76.454,33 €

3. Renta pagadera mientras el asegurado esté con vida, inmediata, anticipada, temporal 20 años, de cuantía inicial 6.000 €, creciendo esta cantidad anualmente en un 5% anual acumulativo.

Solución: 142.276,31 €

- Un asegurado contrata el 31/03/2004 una renta anual y vencida que cobrará a partir de su jubilación y mientras viva. La cuantía a cobrar el primer año en caso de que se haga efectiva asciende a 9.000 €, creciendo esta cantidad anualmente en un 5% del primer término de la renta. Calcular la prima pura única de la operación.

Bases técnicas: tanto de interés técnico 3% efectivo anual; tabla de mortalidad PERM/2000-P.

Fecha de nacimiento del asegurado: 15/06/1964.

Solución: 98.780,12 €

4. Práctica 3: Seguro mixto

4.1. Enunciado

Realizar un programa que calcule la(s) prima(s) pura(s) en el caso en que la prestación de la operación actuarial consista en un seguro mixto (seguro temporal de fallecimiento más capital diferido).

Datos a entrar por pantalla:

1. Respecto al capital diferido:
 - Cuantía del capital en euros
2. Respecto al seguro temporal de fallecimiento:
 - La cuantía siempre constante del seguro podrá ser:
 - Igual a la cuantía del capital diferido
 - Doble de la cuantía del capital diferido
 - Mitad de la cuantía del capital diferido
 - Temporalidad del seguro
3. Respecto a la(s) prima(s) pura(s):
 - Modalidad de prima:
 - Prima única
 - Primas periódicas constantes
 - Temporalidad de las primas (controlando que la misma sea siempre inferior o igual a la temporalidad del seguro)
4. Respecto al asegurado:
 - Fecha de nacimiento
 - Fecha de valoración
5. Bases Técnicas:
 - Tanto de valoración en tanto por uno (se solicita el efectivo anual)
 - Tablas de mortalidad a utilizar: Tablas Generacionales PERM/2000-P

Resultado a presentar por pantalla:

- Valor de la Prima Pura Única o de las Primas Puras Periódicas según la opción elegida

El formulario debe incluir los botones: CALCULAR, OTRO y SALIR.

4.2. Carátula

Sin ejecutar

The screenshot shows a window titled "Seguro Mixto" with a dotted background. It contains several input fields and buttons:

- Capital Diferido:** A text box labeled "Cuantía".
- Datos del Asegurado:** Two date input fields labeled "Fecha de nacimiento (dd/mm/aaaa)" and "Fecha de valoración (dd/mm/aaaa)".
- Seguro de Fallecimiento:** Three radio buttons: "Misma cuantía del capital diferido", "Doble cuantía del capital diferido", and "Mitad de la cuantía del capital diferido". Below them is a "Temporalidad" text box.
- Bases Técnicas:** An "Interés Efectivo Anual (en tanto por uno)" text box and a dropdown menu for "Tablas Generacionales PERM/2000-P" with navigation arrows.
- Primas:** Two radio buttons: "Prima única" and "Primas periódicas". Below them is a "Temporalidad" text box.
- Output area:** Two lines of text: "Valor de la Prima Pura Única en euros" and "Valor de la Prima Periódica en euros", followed by a large empty text box.
- Buttons:** "CALCULAR", "OTRO", and "SALIR" at the bottom.

Ejecutada

The screenshot shows the same "Seguro Mixto" window after calculation. The layout is identical to the previous state, but with the following changes:

- The "CALCULAR" button is now highlighted with a dashed border, indicating it has been clicked.
- The "Valor de la Prima Pura Única en euros" and "Valor de la Prima Periódica en euros" text boxes now contain numerical results.
- A mouse cursor is visible over the empty text box below the output labels.

4.3. Controles

CONTROL	(NOMBRE)	CAPTION	TEXT
Formulario	frmPráctica3	Seguro Mixto	
Frame	frCapitalDiferido	Capital Diferido	
Frame	frSeguroFallecimiento	Seguro de Fallecimiento	
Frame	frPrimas	Primas	
Frame	frDatosAsegurado	Datos del Asegurado	
Frame	frBasesTecnicas	Bases Técnicas	
Label	lblCuantia	Cuantía	
Text Box	txtCuantia		En blanco
Option Button	opbIgual	Misma cuantía del capital diferido	
Option Button	opbDoble	Doble cuantía del capital diferido	
Option Button	opbMitad	Mitad de la cuantía del capital diferido	
Label	lblTemporalidadSeguro	Temporalidad	
Text Box	txtTemporalidadSeguro		En blanco
Option Button	opbUnica	Prima única	
Option Button	opbPeriodicas	Primas periódicas	

Label	lblTemporalidadPrimas	Temporalidad	
Text Box	txtTemporalidadPrimas		En blanco
Label	lblFechaNacimiento	Fecha de Nacimiento (dd/mm/aaaa)	
Text Box	txtFechaNacimiento		En blanco
Label	lblFechaValoracion	Fecha de Valoración (dd/mm/aaaa)	
Text Box	txtFechaValoracion		En blanco
Label	lblInteres	Interés Efectivo Anual (en tanto por uno)	
Text Box	txtInteres		En blanco
Label	lblTablas	Tablas Generacionales PERM/2000-P	
Data	dataTablas	Tablas Generacionales	
Command Button	cmdCALCULAR	CALCULAR	
Command Button	cmdOTRO	OTRO	
Command Button	cmdSALIR	SALIR	
Label	lblValorPrimaUnica	Valor de la Prima Única en euros	
Label	lblValorPrimaPeriodica	Valor de la Prima Periódica en euros	
Label	lblResultado	En blanco	

4.4. Código

'Definición de Variables en (General) (Declaraciones)

```
Dim Interes As Double
Dim NuevolInteres As String
Dim NuevaCuantia As String
Dim NuevaTemporalidadPrimas As String
Dim TemporalidadSeguro As Integer
Dim TemporalidadPrimas As Integer
Dim Cuantia As Double
Dim PrimaUnica As Double
Dim PrimaPeriodica As Double
Dim Prima As Double
Dim ValorActualPrimas As Double
Dim Prestaciones As Double
Dim CapitalDiferido As Double
Dim SeguroUnitario As Double
Dim EdadActuarial As Integer
Dim f1 As Date
Dim f2 As Date
Dim L(200) As Double
Dim AñoNacimiento As Integer
```

```
Private Sub CmdCalcular_Click()
```

```
'Obtener Información
```

```
    Interes = Val(txtInteres.Text)
    If Interes < 0 Then
        NuevolInteres = InputBox("El Interés debe ser estrictamente positivo." & vbCrLf & "Introduzca de nuevo el Interés.", "ERROR")
        If NuevolInteres = "" Then
            cmdOTRO_Click
            Exit Sub
        End If
        Interes = Val(NuevolInteres)
        txtInteres.Text = Str(Interes)
    End If
    TemporalidadSeguro = Val(txtTemporalidadSeguro.Text)
    TemporalidadPrimas = Val(txtTemporalidadPrimas.Text)
    If TemporalidadPrimas > TemporalidadSeguro Then
        NuevaTemporalidadPrimas = InputBox("La temporalidad de las primas debe ser <= a la temporalidad del seguro" & vbCrLf & "Introduzca de nuevo la temporalidad de las primas.", "ERROR")
        If NuevaTemporalidadPrimas = "" Then
            cmdOTRO_Click
            Exit Sub
        End If
        TemporalidadPrimas = Val(NuevaTemporalidadPrimas)
        txtTemporalidadPrimas.Text = Str(TemporalidadPrimas)
    End If
    Cuantia = Val(txtCuantia.Text)
    If Cuantia < 0 Then
        NuevaCuantia = InputBox("La Cuantía debe ser positiva" & vbCrLf & "Introduzca de nuevo la Cuantía.", "ERROR")
        If NuevaCuantia = "" Then
            cmdOTRO_Click
            Exit Sub
        End If
    End If
```

```
End If
Cuantia = Val(NuevaCuantia)
txtCuantia.Text = Str(Cuantia)
End If
f1 = txtFechaNacimiento.Text
If Year(f1) < 1885 Or Year(f1) > 2001 Then
    Res = MsgBox("La Fecha de nacimiento es incorrecta." & vbCrLf & "Introduzca de nuevo la fecha.",
vbOKCancel + vbExclamation + vbDefaultButton1, "ERROR")
    If Res = vbOK Then
        txtFechaNacimiento.Text = " "
        txtFechaNacimiento.SetFocus
        Exit Sub
    Else
        cmdOTRO_Click
        Exit Sub
    End If
End If
f2 = txtFechaValoracion.Text
If Year(f2) <= Year(f1) Then
    Res = MsgBox("La Fecha de valoración es incorrecta." & vbCrLf & "Introduzca de nuevo la fecha.",
vbOKCancel + vbExclamation + vbDefaultButton1, "ERROR")
    If Res = vbOK Then
        txtFechaValoracion.Text = " "
        txtFechaValoracion.SetFocus
        Exit Sub
    Else
        cmdOTRO_Click
        Exit Sub
    End If
End If

'Cálculo de la edad Actuarial

EdadActuarial = Int(((f2 - f1) / 365.25) + 0.5)

'Lectura de la tabla generacional

AñoNacimiento = Year(f1)
campo = Trim(AñoNacimiento)
dataTablas.Recordset.MoveFirst
For i = 0 To 116
    L(i) = dataTablas.Recordset.Fields(campo)
    dataTablas.Recordset.MoveNext
Next i

'Cálculo de la Prima

'Calculo del Capital diferido

CapitalDiferido = Cuantia * ((1 + Interes) ^ (-TemporalidadSeguro)) * L(EdadActuarial + TemporalidadSeguro) /
L(EdadActuarial)

'Calculo del Seguro Unitario

SeguroUnitario = 0

For t = 0 To TemporalidadSeguro - 1
```

```
SeguroUnitario = SeguroUnitario + ((1 + Interes) ^ -(t + 1)) * (L(EdadActuarial + t) - L(EdadActuarial + t + 1)) / L(EdadActuarial)
Next t
```

```
'Calculo del Valor Actual de las Prestaciones
```

```
If opbIguar.Value = True Then
    ValorActualPrestaciones = CapitalDiferido + Cuantia * SeguroUnitario
ElseIf opbDoble.Value = True Then
    ValorActualPrestaciones = CapitalDiferido + 2 * Cuantia * SeguroUnitario
ElseIf opbMitad.Value = True Then
    ValorActualPrestaciones = CapitalDiferido + (1 / 2) * Cuantia * SeguroUnitario
End If
```

```
'Cálculo del Valor Actual de las Primas
```

```
If opbPeriodicas.Value = True Then
    ValorActualPrimas = 0
    For s = 0 To TemporalidadPrimas - 1
        ValorActualPrimas = ValorActualPrimas + ((1 + Interes) ^ (-s)) * L(EdadActuarial + s) / L(EdadActuarial)
    Next s
    PrimaPeriodica = ValorActualPrestaciones / ValorActualPrimas
    Prima = PrimaPeriodica
Else
    PrimaUnica = ValorActualPrestaciones
    Prima = PrimaUnica
End If
```

```
IblResultado.Caption = Str(Prima)
```

```
End Sub
```

```
Private Sub cmdOTRO_Click()
    txtCuantia.Text = " "
    txtInteres.Text = " "
    opbIguar.Value = False
    opbDoble.Value = False
    opbMitad.Value = False
    txtTemporalidadSeguro.Text = " "
    opbUnica.Value = False
    opbPeriodicas.Value = False
    txtTemporalidadPrimas.Text = " "
    txtFechaNacimiento = " "
    txtFechaValoracion = " "
    IblValorPrimaUnica.Visible = False
    IblValorPrimaPeriodica.Visible = False
    IblResultado.Caption = " "
    IblResultado.Visible = False
    IblTemporalidadPrimas.Visible = False
    txtTemporalidadPrimas.Visible = False
    txtCuantia.SetFocus
End Sub
```

```
Private Sub CmdSalir_Click()
    End 'Sale del programa
```

End Sub

```
Private Sub opbUnica_Click()  
    lblValorPrimaUnica.Visible = True  
    lblResultado.Visible = True  
    lblValorPrimaPeriodica.Visible = False  
    lblTemporalidadPrimas.Visible = False  
    txtTemporalidadPrimas.Visible = False  
End Sub
```

```
Private Sub opbPeriodicas_Click()  
    lblValorPrimaUnica.Visible = False  
    lblResultado.Visible = True  
    lblValorPrimaPeriodica.Visible = True  
    lblTemporalidadPrimas.Visible = True  
    txtTemporalidadPrimas.Visible = True  
End Sub
```

4.5. Ejemplos numéricos

- Una persona contrata el 01/03/2004 un seguro mixto de forma que si llega vivo a los 65 años de edad cobrará 12.000 € y la misma cantidad cobrarán sus herederos al final del año de su fallecimiento si este tiene lugar durante los próximos 15 años. Calcular la cuantía de la prima única a abonar por el asegurado.

Bases técnicas: interés de valoración 3% efectivo anual; tabla de mortalidad PERM/2000-P.

Fecha de nacimiento del asegurado: 05/05/1954.

Solución: 7.824,77€

- Una persona contrata el 01/03/2004 un seguro mixto por el que si llega vivo a los 65 años de edad cobrará 12.000 € o si por el contrario fallece durante los próximos 25 años, sus herederos cobrarán 24.000 €. Calcular el importe de la prima a abonar por el asegurado en los siguientes casos:

1. Prima única.
2. Primas constantes, anuales, pagaderas mientras viva el asegurado, y como máximo durante 10 años.

Bases técnicas: interés de valoración 3% efectivo anual; tabla de mortalidad PERM/2000-P.

Fecha de nacimiento del asegurado: 25/02/1964.

Solución:

Prima única: 6.539,09 €

Prima periódica: 749,35 €

- Una persona contrata el 01/03/2004 un seguro mixto por el que si llega vivo a los 65 años de edad cobrará 18.000 € y sus herederos cobrarán la mitad de esta cantidad al final del año de su fallecimiento si éste tiene lugar durante los próximos 30 años. Calcular el importe de la prima a abonar por el asegurado en los siguientes casos:

1. Prima única.
2. Primas constantes, anuales, pagaderas mientras viva el asegurado, y como máximo durante 15 años.

Bases técnicas: interés de valoración 3% efectivo anual; tabla de mortalidad PERM/2000-P.

Fecha de nacimiento del asegurado: 30/04/1969.

Solución:

Prima única: 7.215,33 €

Prima periódica: 591,66 €